
1 Introduction

Nitpick is a new counterexample generator for Isabelle/HOL [5] that is designed
to handle formulas combining inductive datatypes, inductively defined predi-
cates, and quantifiers. It builds on Kodkod [7], a highly optimized first-order
relational model finder developed by the Software Design Group at the MIT.
It is conceptually similar to Refute [8], from which it borrows many ideas and
code fragments, but it benefits from Kodkod’s optimizations and a new encod-
ing scheme. The name Nitpick is shamelessly stolen from a (now obsolete) tool
developed in the 1990s by the Software Design Group.

Nitpick is easy to use—you simply enter nitpick after a putative theorem and
wait a few seconds. Nonetheless, there are situations where knowing how it
works behind the scenes and how it reacts to various options helps increase the
test coverage. This manual also explains how to install the tool on your work-
station. Should the motivation fail you, think of the many hours of hard work
Nitpick will save you. Proving non-theorems is hard work.

2 Installing the Tool

To install Nitpick, download and extract the archive http://isabelle.in.tum.
de/~blanchet/Nitpick-1.0.0.tgz, then run the build script. The script per-
forms the following steps:

1. It installs the Kodkod library, the Kodkodi front-end, and the portable SAT4J
solver in the $ISABELLE_HOME/contrib/kodkodi directory.

2. It builds a HOL-Nitpick Isabelle image in the $ISABELLE_OUTPUT directory.

3. It builds an associated keyword file isar-keywords-HOL-Nitpick.el in the
$ISABELLE_HOME_USER/etc directory.

To activate Nitpick, you must either invoke isabelle emacs with the option -L
HOL-Nitpick or choose the HOL-Nitpick logic from the Isabelle menu in Emacs.

3 First Steps

This section introduces Nitpick by presenting small examples. If possible, you
should try out the examples on your workstation. Your theory file should start as
follows:

theory Scratch
imports Main Nitpick
begin

2



To obtain the same results as presented here, make sure that SAT4J is used as the
SAT solver by adding the line

nitpick_params [sat_solver = SAT4J]

after the begin keyword. Being written in Java, SAT4J is the most portable SAT
solver supported by Nitpick and the only one that is guaranteed to be available
if you followed the instructions in §2. Faster SAT solvers can also be installed,
as explained in §5.5. If you already configured SAT solvers in Isabelle (e.g., for
Refute), these will also be available to Nitpick.

3.1 Propositional Logic

Let’s start with a trivial example from propositional logic:

lemma “P −→ Q”
nitpick

If Nitpick is correctly installed, you should get the following output:

Nitpick found a genuine counterexample:

Free variables:
P = True
Q = False

In cases like this, there is only one thing to say:

oops

3.2 Type Variables

If you are left unimpressed by the previous example, don’t worry. The next one
is more mind- and computer-boggling:

lemma “P x =⇒ P (THE y. P y)”

The putative lemma involves the definite description operator, THE, presented
in section 5.10.1 of the Isabelle tutorial [5]. The operator is defined by the axiom
(THE x. x = a) = a. The putative lemma is merely asserting the indefinite
description operator axiom with THE substituted for SOME.

The free variable x and the bound variable y have type ′a. For formulas contain-
ing type variables, Nitpick enumerates the possible domains for each type vari-
able, up to a given cardinality (6 by default), looking for a finite countermodel:

nitpick [verbose = true]

Trying 6 scopes:
card ′a = 1;
card ′a = 2;
card ′a = 3;
card ′a = 4;
card ′a = 5;
card ′a = 6.

Nitpick found a genuine counterexample for card ′a = 3:

Free variables:
P = {a2, a3}
x = a3

Nitpick found a counterexample in which ′a has cardinality 3. (For cardinalities
1 and 2, the formula holds.) In the counterexample, the three values of type ′a are
written a1, a2, and a3.

The message “Trying n scopes: . . . ” is shown only if the option verbose is set to
true. You can specify verbose = true each time you invoke nitpick, or you can set
it globally using the command

nitpick_params [verbose = true]

The command also displays the current default values for all of the options sup-
ported by Nitpick. The options are listed in §5.

3.3 Constants

By just looking at Nitpick’s output, it might not be clear why the counterexample
in §3.2 is genuine. Let’s invoke Nitpick again, this time telling it to show the
values of the constants that occur in the formula:

lemma “P x =⇒ P (THE y. P y)”
nitpick [show_consts = true]

Nitpick found a genuine counterexample for card ′a = 3:

Free variables:
P = {a2, a3}
x = a3

Constant:
THE y. P y = a1

We can see more clearly now. Since the predicate P isn’t true for a unique value,
THE y. P y can denote any value of type ′a, even a1. Since P a1 is false, the entire
formula is falsified.

As an optimization, Nitpick directly assigned a value to the expression THE y. P y
(i.e., The (λy. P y)), rather than to the The constant. If we disable this optimization
by using the command

nitpick [specialize_funs = false, show_consts = true]

we get The as expected:

Constant:
The = undefined({} := a3, {a3} := a3, {a2} := a2,

{a2, a3} := a1, {a1} := a1, {a1, a3} := a3,
{a1, a2} := a3, {a1, a2, a3} := a3)

Notice that The (λy. P y) = The {a2, a3} = a1, just like before.1

Our misadventures with THE suggest adding ‘∃!x.’ (“there exists a unique x such
that”) at the front of our putative lemma’s assumption:

lemma “∃!x. P x =⇒ P (THE y. P y)”

The fix appears to work:

nitpick

Nitpick found no counterexample.

We can further increase our confidence in the formula by exhausting all cardinal-
ities up to 100:

nitpick [card ′a = 1–100]

Nitpick found no counterexample.

Let’s see if Sledgehammer [6] can find a proof:

sledgehammer

Sledgehammer: external prover “e” for subgoal 1:
∃!x. P x =⇒ P (THE y. P y)
Try this command: apply (metis the_equality)

apply (metis the_equality)

No subgoals!

This must be our lucky day.

1The undefined symbol’s presence is explained as follows: In higher-order logic, any function
can be built from the undefined function using repeated applications of the function update op-
erator f (x := y), just like any list can be built from the empty list using z # zs.

3.4 Skolemization

Are all invertible functions onto? Let’s find out:

lemma “∃g. ∀x. g ( f x) = x =⇒ ∀y. ∃x. y = f x”
nitpick

Nitpick found a genuine counterexample for card ′a = 2 and card ′b = 1:

Free variable:
f = undefined(b1 := a1)

Skolem constants:
g = undefined(a1 := b1, a2 := b1)
y = a2

Although f is the only free variable occurring in the formula, Nitpick also dis-
plays values for the bound variables g and y. These values are available to Nit-
pick because it performs skolemization as a preprocessing step.

In the previous example, skolemization only affected the outermost quantifiers.
This is not always the case, as illustrated below:

lemma “∃x. ∀ f . f x = x”
nitpick

Nitpick found a genuine counterexample for card ′a = 2:

Skolem constant:
λx. f = undefined(a1 := undefined(a1 := a2, a2 := a1),

a2 := undefined(a1 := a1, a2 := a1))

The variable f is bound within the scope of x; therefore, f depends on x, as sug-
gested by the notation λx. f . If x = a1, then f is the function that maps a1 to a2
and vice versa; otherwise, x = a2 and f maps both a1 and a2 to a1. In both cases,
f x 6= x.

The source of the Skolem constants is sometimes more obscure:

lemma “reflexive r =⇒ sym r”
nitpick

Nitpick found a genuine counterexample:

Free variable:
r = {(a1, a1), (a2, a1), (a2, a2)}

Skolem constants:
sym.x = a2
sym.y = a1

What happened here is that Nitpick expanded the sym constant to its definition:

sym r ≡ ∀x y. (x, y) ∈ r −→ (x, y) ∈ r.

As their names suggest, the Skolem constants sym.x and sym.y are simply the
bound variables x and y from sym’s definition.

Although skolemization is a useful optimization, you can disable it by invoking
Nitpick with skolem_depth = −1. See §5.5 for details.

3.5 Natural Numbers and Integers

Because of the axiom of infinity, the type nat does not admit any finite models.
To work around this, Nitpick considers prefixes {0, 1, . . . , K − 1} of nat (where
K = card nat) and maps all other numbers to the undefined value (⊥). The type
int is handled in a similar way: If K = card int, the fragment of int known to
Nitpick is {−dK/2e+ 1, . . . , +bK/2c}.

Undefined values lead to a three-valued logic. If the formula evaluates to ⊥,
Nitpick reports the model as a potential counterexample and continues looking
for a model that makes the negated formula evaluate to True. For example:

lemma “[[i ≤ j; n ≤ (m::nat)]] =⇒ i ∗ n + j ∗m ≤ i ∗m + j ∗ n”
nitpick

Nitpick found a potential counterexample for card nat = 2:

Free variables:
i = 1
j = 1
m = 1
n = 1

Nitpick found a genuine counterexample for card nat = 2:

Free variables:
i = 0
j = 1
m = 1
n = 0

Nitpick first finds a potential counterexample for card nat = 2, which happens to
be spurious; then it finds a genuine one for the same cardinality. The spurious
counterexample manifests itself because 1 ∗ 1 + 1 ∗ 1 equals 2, which lies outside
the fragment {0, 1}: The right-hand side of =⇒ then evaluates to ⊥, and the left-
hand side evaluates to True; following Kleene’s three-valued logic, this gives ⊥
for the entire formula.

One way to carry out such investigations is to install a watch on the subterms of
interest using the watch option:

nitpick [watch = “i ∗ n + j ∗m; i ∗m + j ∗ n”]

Nitpick found a potential counterexample for card nat = 2:

Free variables:
i = 1
j = 1
m = 1
n = 1

Watched terms:
i ∗ n + j ∗m = ⊥
i ∗m + j ∗ n = ⊥

Nitpick found a genuine counterexample for card nat = 2:

Free variables:
i = 0
j = 1
m = 1
n = 0

Watched terms:
i ∗ n + j ∗m = 0
i ∗m + j ∗ n = 1

We don’t always have the luxury of a genuine counterexample and must often
content ourselves with a potential one. The tedious task of finding out whether
the potential counterexample is in fact genuine can be outsourced to auto by pass-
ing the option check_potential = true. For example:

lemma “∀n. Suc n > n =⇒ P”
nitpick [card nat = 100, check_potential = true]

Nitpick found a potential counterexample for card nat = 100:

Free variable:
P = False

Confirmation by “auto”: The above counterexample is genuine.

You might wonder why the counterexample is first reported as potential. The
root of the problem is that n in ∀n. Suc n > n ranges over an infinite type. If
Nitpick finds an n such that Suc n 6> n, it evaluates the assumption to False; but
otherwise, it does not know anything about values of n ≥ card nat and must
therefore evaluate the assumption to ⊥, not True. Since the assumption can never
be satisfied, the putative lemma can never be falsified.

If you distrust the so-called genuine counterexamples, you can set check_genuine =
true to verify them as well. However, be aware that auto will often fail to prove
that the counterexample is genuine or spurious.

A final remark before we move on: Because numbers are infinite and are approx-
imated using a three-valued logic, there is no need to systematically enumerate
domain sizes. If Nitpick cannot find a genuine counterexample for card nat = 10,
none could be found for smaller domains. Nitpick nonetheless enumerates all

cardinalities from 1 to 6 for nat, because smaller cardinalities are fast to handle
and give rise to simpler counterexamples.

3.6 Inductive Datatypes

Like natural numbers and integers, inductive datatypes with recursive construc-
tors admit no finite models and must be approximated. Conceptually, each con-
structor is seen as building a different kind of object, for which you can specify a
multiplicity. For example, using a multiplicity of 1 for Nil and 10 for Cons, Nitpick
looks for all counterexamples that can be built using at most 11 different lists.

Let’s see with an example involving hd (which returns the first element of a list)
and @ (which concatenates two lists):

lemma “hd (xs @ ys) = hd xs”
nitpick

Nitpick found a potential counterexample for card ′a = 1, mult Nil = 1, and
mult Cons = 1:

Free variables:
xs = [a1]
ys = [a1]

Nitpick found a genuine counterexample for card ′a = 2, mult Nil = 1, and
mult Cons = 1:

Free variables:
xs = []
ys = [a1]

To see why the second counterexample is genuine, we set show_consts = true and
show_datatypes = true:

Datatype:
′a list = {[], [a1], ⊥}

Constants:
hd = undefined([] := a2, [a1] := a1)
? @ ys = undefined([] := [a1], [a1] := ⊥)

Since hd [] is undefined in the logic, it may be given any value, including a1.

The second constant, ? @ ys, is simply the append operator whose second argu-
ment is fixed to be ys (i.e., λxs. xs @ ys). Appending [a1] to itself would normally
give [a1, a1], but this value is not representable in the fragment of ′a list considered
by Nitpick, which is shown under the “Datatype” heading.

Given card ′a = 2, mult Nil = 1, and mult Cons = 2, Nitpick considers the follow-
ing fragments:

{[], [a1], [a2]};
{[], [a2], [a1, a2]};

{[], [a1], [a1, a1]};
{[], [a2], [a2, a2]}.

{[], [a1], [a2, a1]};

All subterm-closed fragments consisting of one Nil and two Conses are listed and
only those. As an example of a non-subterm-closed fragment, consider F =
{[], [a1], [a1, a2]}, and observe that [a1, a2] (i.e., Cons a1 [a2]) has [a2] /∈ F as a
subterm.

Here’s another möchtegern-lemma that Nitpick can refute without a blink:

lemma “[x] = [y]”
nitpick [show_datatypes = true, max_potential = 0]

Nitpick found a genuine counterexample for card ′a = 2, mult Nil = 1, and
mult Cons = 2:

Free variables:
x = a2
y = a1

Datatype:
′a list = {[], [a1], [a2],⊥}

This time we asked Nitpick not to display any potential counterexamples by spec-
ifying max_potential = 0.

We saw earlier that type cardinalities range from 1 to 6 by default. Similarly,
Nitpick tries multiplicities 1 to 6 for each constructor. This can be changed using
the mult option, as explained in §5.2. Because datatypes are approximated using
a three-valued logic, there is no need to systematically enumerate multiplicities:
If Nitpick cannot find a genuine counterexample for mult Cons = 10, none could
be found for smaller multiplicities.

Inconsistencies in the multiplicity settings are automatically resolved. For exam-
ple, setting mult Nil = 2 will result in a multiplicity of 1.

3.7 Typedefs and Records

Nitpick generally treats types declared using typedef as datatypes whose single
constructor is the corresponding Abs_ function. For example:

typedef three = “{0::nat, 1, 2}”
by blast

definition A :: three where “A ≡ Abs_three 0”
definition B :: three where “B ≡ Abs_three 1”
definition C :: three where “C ≡ Abs_three 2”

lemma “[[P A; P B]] =⇒ P x”
nitpick [show_datatypes = true]

Nitpick found a genuine counterexample for card nat = 3 and mult Abs_three
= 3:

Free variables:
P = {Abs_three 0, Abs_three 1}
x = Abs_three 2

Datatypes:
nat = {0, 1, 2,⊥}
three = {Abs_three 0, Abs_three 1, Abs_three 2, ⊥}

Records, which are implemented as typedefs behind the scenes, are handled in
exactly the same way:

record point =
Xcoord :: int
Ycoord :: int

lemma “Xcoord (|Xcoord = x, Ycoord = y|) = y”
nitpick [max_potential = 0]

Nitpick found a genuine counterexample for card int = 2 and mult Abs_point_ext_type
= 2:

Free variables:
x = 1
y = 0

3.8 Inductively Defined Predicates

Inductively defined predicates (and sets) are particularly problematic for coun-
terexample generators. They can make Quickcheck [2] loop endlessly and Refute
[8] run out of resources. The crux of the problem is that they are defined using an
expensive least fixed point construction.

Nitpick addresses this problem by first acknowledging that not all inductively
defined predicates are equal. Consider the even predicate below:

inductive even where
“even 0” |
“even n =⇒ even (Suc (Suc n))”

This predicate enjoys the desirable property of being wellfounded, which means
that the introduction rules don’t give rise to infinite chains of the form

· · · =⇒ even k′′ =⇒ even k′ =⇒ even k.

For even, this is obvious: Any chain ending at k will be of length k/2 + 1:

even 0 =⇒ even 2 =⇒ · · · =⇒ even (k− 2) =⇒ even k.

Wellfoundedness is desirable because it enables Nitpick to use a very efficient
fixed point computation.2 Moreover, Nitpick can prove wellfoundedness of most
wellfounded predicates, just as Isabelle’s function package can usually discharge
termination proof obligations automatically.

Let’s try an example:

lemma “∃n. even n ∧ even (Suc n)”
nitpick [card nat = 100, verbose = true, show_consts = true]

The inductively defined predicate “even” was proved wellfounded. Nitpick
can compute it efficiently.

Trying 1 scope:
card nat = 100.

Nitpick found a potential counterexample for card nat = 100:

Constant:
even = {0, 2, 4, 6, 8, 10, 12, 14, 16, 18, 20, 22, 24, 26, 28, 30, 32, 34,

36, 38, 40, 42, 44, 46, 48, 50, 52, 54, 56, 58, 60, 62, 64, 66,
68, 70, 72, 74, 76, 78, 80, 82, 84, 86, 88, 90, 92, 94, 96, 98}

Nitpick could not find a better counterexample.

We see that Nitpick correctly computed the set of even numbers less than 100.
Nitpick cannot rule out the existence of a natural number n ≥ 100 such that both
even n and even (Suc n) are true. To help Nitpick, we could bound the existential
quantifier:

lemma “∃n≤ 99. even n ∧ even (Suc n)”
nitpick [card nat = 100]

Nitpick found a genuine counterexample for card nat = 100:

Trivial

So far we were blessed by the wellfoundedness of even. What happens if we use
the following definition instead?

inductive even′ where
“even′ (0::nat)” |
“even’ 2” |
“[[even′ m; even′ n]] =⇒ even′ (m + n)”

This definition is not wellfounded: From even′ 0 and even′ 0, we can derive that
even′ 0. Nonetheless, the predicates even and even′ are equivalent.

2If an inductive predicate is wellfounded, then it has exactly one fixed point, which is simul-
taneously the least and the greatest fixed point. In these circumstances, the computation of the
least fixed point amounts to the computation of an arbitrary fixed point, which can be performed
using a straightforward recursive equation.

12



Let’s check a property involving even′. To make up for the foreseeable computa-
tional hurdles entailed by non-wellfoundedness, we decrease nat’s cardinality to
a mere 10:

lemma “¬ even′ 8”
nitpick [card nat = 10, verbose = true, show_consts = true]

The inductively defined predicate “even′” could not be proved wellfounded.
Nitpick will unroll the predicate.

Trying 5 scopes:
card nat = 10 and iter even′ = 1;
card nat = 10 and iter even′ = 2;
card nat = 10 and iter even′ = 4;
card nat = 10 and iter even′ = 8;
card nat = 10 and iter even′ = 9.

Nitpick found a potential counterexample for card nat = 10 and iter even′ = 1:

Constant:
λi. even′ = undefined(1 := {0, 2, 4, 1?, 3?, 5?, 6?, 7?, 8?, 9?},

0 := {0, 2, 1?, 3?, 4?, 5?, 6?, 7?, 8?, 9?})
Nitpick found a genuine counterexample for card nat = 10 and iter even′ = 2:

Constant:
λi. even′ = undefined(2 := {0, 2, 4, 6, 8, 1?, 3?, 5?, 7?, 9?},

1 := {0, 2, 4, 1?, 3?, 5?, 6?, 7?, 8?, 9?},
0 := {0, 2, 1?, 3?, 4?, 5?, 6?, 7?, 8?, 9?})

Nitpick’s output is very instructive. First, it tells us that the predicate is unrolled,
meaning that it is computed iteratively from the empty set. Then it lists five
scopes specifying different bounds on the numbers of iterations: 1, 2, 4, 8, and 9.

The output also shows how each iteration contributes to even′. The notation
λi. even′ indicates that the value of the predicate depends on an iteration counter.
Iteration 0 provides the basis elements, 0 and 2. Iteration 1 contributes 4 (= 2 + 2).
Iteration 2 throws 6 (= 2 + 4 = 4 + 2) and 8 (= 4 + 4) into the mix. Further itera-
tions would not contribute any new elements.

Some values are marked with subscripted question marks (‘?’). These are the
elements for which the predicate evaluates to ⊥. Thus, even′ evaluates to either
True or ⊥, never False. This reflects the fundamental incompleteness of predicate
unrolling.

When unrolling a predicate, Nitpick tries 1, 2, 4, 8, 16, and 32 iterations. However,
these numbers are bounded by the cardinality of the predicate’s domain. With
card nat = 10, no more than 9 iterations are ever needed to compute the value of
a nat predicate. You can specify the number of iterations using the iter option, as
explained in §5.2.

4 Case Studies

As a didactic device, the previous section focused on toy formulas whose validity
can easily be assessed just by looking at the formula. We will now two somewhat
more realistic case studies that are within Nitpick’s reach: a context-free grammar
modeled by mutually inductive sets and a functional implementation of AA trees.

The examples being larger, more time is spent solving SAT problems. If you
haven’t done so already, now would be a perfect time to install MiniSat 1.14.
To reproduce the results presented in this section, add the command

nitpick_params [max_potential = 0, sat_solver = MiniSat]

at the beginning of your theory.

4.1 A Context-Free Grammar

Our first case study is taken from section 7.4 in the Isabelle tutorial [5]. The fol-
lowing grammar, originally due to Hopcroft and Ullman, produces all strings
with an equal number of a’s and b’s:

S ::= ε | bA | aB
A ::= aS | bAA
B ::= bS | aBB

The intuition behind the grammar is that A generates all string with one more a
than b’s and B generates all strings with one more b than a’s.

The alphabet consists exclusively of a’s and b’s:

datatype alphabet = a | b

Strings over the alphabet are represented by alphabet lists. Nonterminals in the
grammar become sets of strings. The production rules presented above can be
expressed as a mutually inductive definition:

inductive_set S and A and B where
R1: “[] ∈ S” |
R2: “w ∈ A =⇒ b # w ∈ S” |
R3: “w ∈ B =⇒ a # w ∈ S” |
R4: “w ∈ S =⇒ a # w ∈ A” |
R5: “w ∈ S =⇒ b # w ∈ S” |
R6: “[[v ∈ B; v ∈ B]] =⇒ a # v @ w ∈ B”

The conversion of the grammar into the inductive definition was done manually
by Joe Blow, an underpaid undergraduate student. As a result, some errors might
have sneaked in.

Debugging faulty specifications is part of Nitpick’s raison d’être. A good ap-
proach is to state desirable properties of the specification (here, that S is the ex-
actly set of strings over {a, b} with as many a’s as b’s) and check them with Nit-
pick. If the properties are correctly stated, counterexamples point to bugs in the
specification. For our grammar example, we will proceed in two step, separating
the soundness and the completeness of the set S. First, soundness:

theorem S_sound:
“w ∈ S −→ length [x← w. x = a] = length [x← w. x = b]”
nitpick

Nitpick found a genuine counterexample for card nat = 2, mult Nil = 1, mult
Cons = 2, mult a = 1, and mult b = 1:

Free variable:
w = [b]

It would seem that [b] ∈ S. How could this be? An inspection of the introduction
rules reveals that the only rule with a right-hand side of the form b # . . . ∈ S that
could have introduced [b] into S is R5:

“w ∈ S =⇒ b # w ∈ S”

On closer inspection, this rule is wrong. To match the production B ::= bS, the
second S should be a B. We fix the typo and try again:

nitpick

Nitpick found a genuine counterexample for card nat = 5, mult Nil = 1, mult
Cons = 5, mult a = 1, and mult b = 1:

Free variable:
w = [a, a, b]

Some detective work is necessary to find out what went wrong. To get [a, a, b] ∈ S,
we need [a, b] ∈ B by R3, which in turn can only come from R6:

“[[v ∈ B; v ∈ B]] =⇒ a # v @ w ∈ B”

Now, this formula must be wrong: The same assumption occurs twice, and the
variable w is unconstrained. Clearly, either of the two occurrences of v in the
assumptions should have been a w.

If we try again, we don’t get any counterexample from Nitpick, even if we enlarge
the scope. Let’s move on and check completeness:

theorem S_complete:
“length [x← w. x = a] = length [x← w. x = b] −→ w ∈ S”
nitpick

Nitpick found a genuine counterexample for card nat = 6, mult Nil = 1, mult
Cons = 6, mult a = 1, and mult b = 1:

Free variable:
w = [b, b, a, a]

Apparently, [b, b, a, a] /∈ S, even though it has the same numbers of a’s and b’s. But
since our inductive definition passed the soundness check, the introduction rules
we have are probably correct. Perhaps we simply lack an introduction rule. Com-
paring the grammar with the inductive definition, our suspicion is confirmed: Joe
Blow simply forgot the production A ::= bAA, without which the grammar can-
not generate two or more b’s in a row. So we add the introduction rule

“[[v ∈ A; w ∈ A]] =⇒ b # v @ w ∈ A”

With this last change, we don’t get any counterexamples from Nitpick for either
soundness or completeness. We can even generalize our result to cover A and B
as well:

theorem S_A_B_sound_and_complete:
“w ∈ S↔ length [x← w. x = a] = length [x← w. x = b]”
“w ∈ A↔ length [x← w. x = a] = length [x← w. x = b] + 1”
“w ∈ B↔ length [x← w. x = b] = length [x← w. x = a] + 1”
nitpick [timeout = 60 s]

Nitpick found no counterexample.

The timeout option is given to override Nitpick’s default timeout delay, which is
set to 30 seconds.

4.2 AA Trees

AA trees are a kind of balanced trees introduced by Arne Andersson that provide
similar performance to red-black trees but a simpler implementation [1]. They
can be used to store sets of elements equipped with a total order <. We start by
defining the datatype and some basic extractor functions:

datatype ′a tree = Λ | N “′a::linorder” nat “′a tree” “′a tree”

primrec data where
“data Λ = undefined” |
“data (N x _ _ _) = x”

primrec dataset where
“dataset Λ = {}” |
“dataset (N x _ t u) = {x} ∪ dataset t ∪ dataset u”

primrec level where
“level Λ = 0” |
“level (N _ k _ _) = k”

primrec left where
“left Λ = Λ” |

“left (N _ _ t _) = t”

primrec right where
“right Λ = Λ” |
“right (N _ _ _ u) = u”

The wellformedness criterion for AA trees is fairly complex. To quote every-
body’s favorite source of information, Wikipedia [9]:

Each node has a level field, and the following invariants must remain
true for the tree to be valid:

1. The level of a leaf node is one.

2. The level of a left child is strictly less than that of its parent.

3. The level of a right child is less than or equal to that of its parent.

4. The level of a right grandchild is strictly less than that of its
grandparent.

5. Every node of level greater than one must have two children.

The wf predicate formalizes this description:

primrec wf where
“wf Λ = True” |
“wf (N _ k t u) =
(if t = Λ then

k = 1 ∧ (u = Λ ∨ (level u = 1 ∧ left u = Λ ∧ right u = Λ))
else

wf t ∧ wf u ∧ u 6= Λ ∧ level t < k ∧ level u ≤ k ∧ level (right u) < k)”

Rebalancing the tree upon insertion and removal of elements is performed by two
auxiliary functions called skew and split, defined below:

primrec skew where
“skew Λ = Λ” |
“skew (N x k t u) =
(if t 6= Λ ∧ k = level t then

N (data t) k (left t) (N x k (right t) u)
else

N x k t u)”

primrec split where
“split Λ = Λ” |
“split (N x k t u) =
(if u 6= Λ ∧ k = level (right u) then

N (data u) (Suc k) (N x k t (left u)) (right u)
else

N x k t u)”

Performing a skew or a split should have no impact on the set of elements stored
in the tree:

theorem dataset_skew_split:
“dataset (skew t) = dataset t”
“dataset (split t) = dataset t”
nitpick

Nitpick ran out of time.

Furthermore, applying skew or split to a wellformed tree should not alter the tree:

theorem wf_skew_split:
“wf t =⇒ skew t = t”
“wf t =⇒ split t = t”
nitpick

Nitpick ran out of time.

Insertion is implemented recursively. It preserves the sort order:

primrec insort where
“insort Λ x = N x 1 Λ Λ” |
“insort (N y k t u) x =
(∗ (split ◦ skew) ∗) (N y k (if x < y then insort t x else t)

(if x > y then insort u x else u))”

Notice that we deliberately commented out the application of skew and split. Let’s
see if this causes any problems:

theorem wf_insort: “wf t =⇒ wf (insort t x)”
nitpick

Nitpick found a genuine counterexample for card nat = 3, card ′a = 2, mult
Λ = 1, and mult N = 3:

Free variables:
t = N a2 1 Λ Λ
x = a1

It’s hard to see why this is a counterexample. To improve readability, we will
restrict the theorem to nat, so that we don’t need to look up the value of the
‘op <’ constant to find out which element is smaller than the other. In addition,
we will install a watch on insort t x. This gives

theorem wf_insort_nat: “wf t =⇒ wf (insort t (x::nat))”
nitpick [watch = “insort t x”]

Nitpick found a genuine counterexample for card nat = 3, mult Λ = 1, and
mult N = 3:

Free variables:
t = N 2 1 Λ Λ

x = 1
Watched term:

insort t x = N 2 1 (N 1 1 Λ Λ) Λ

Nitpick’s output reveals that the element 1 was added as a left child of 2, both
with a level of 1. This violates the second AA tree invariant, which states that a
left child’s level must be less than its parent’s. This shouldn’t come as a surprise,
considering that we commented out the tree rebalancing code. Reintroducing the
code seems to solve the problem:

theorem wf_insort: “wf t =⇒ wf (insort t x)”
nitpick

Nitpick ran out of time.

Insertion should transform the set of elements represented by the tree in the ob-
vious way:

theorem dataset_insort: “dataset (insort t x) = {x} ∪ dataset t”
nitpick

Nitpick found no counterexample.

We could continue like this and sketch a complete theory of AA trees without
performing a single proof. Once the definitions and main theorems are in place
and have been thoroughly tested using Nitpick, we could start working on the
proofs. Developing theories this way saves time, because faulty theorems and
definitions are discovered much earlier in the process.

5 Option Reference

Nitpick’s behavior can be influenced by various options, which can be specified
in brackets after the nitpick command. Default values can be set using nitpick_
params. For example:

nitpick_params [verbose = true, timeout = 60 s]

The options are categorized as follows: mode of operation (§5.1), scope of search
(§5.2), output format (§5.3), automatic counterexample checks (§5.4), optimiza-
tions (§5.5), and timeouts (§5.6). The descriptions below refer to the following
syntactic quantities:

• 〈bool〉: The Boolean values true and false.

• 〈string〉: A string.

• 〈int〉: An integer. Negative integers are prefixed with a hyphen.

• 〈int_range〉: An integer (e.g., 3) or a hyphenated range of integers (e.g., 1–4).

• 〈int_seq〉: A comma-separated sequence of integer ranges (e.g., 1,3,6–8).

• 〈time〉: An integer followed by s (seconds) or ms (milliseconds), or the key-
word none (∞ years).

• 〈const〉: The name of a HOL constant.

• 〈term_list〉: A semicolon-separated list of HOL terms.

• 〈type〉: A HOL type.

Default values are indicated in square brackets.

5.1 Mode of Operation

falsify = 〈bool〉 [true]

Specifies whether Nitpick should look for falsifying examples (counterex-
ample) or satisfying examples (models). This manual assumes throughout
that falsify is set to true.

5.2 Scope of Search

card 〈type〉 = 〈int_seq〉
Specifies the sequence of cardinalities to use for a given type, which may
be a free type (′a, ′b, etc.), a type declared using typedecl (except bool), or
the built-in types nat or int. For nat and int, the cardinality applies to the
fragment used to approximate the type. For example:

card nat = 4 induces the fragment {0, 1, 2, 3}
card int = 4 induces the fragment {−1, 0, +1, +2}
card int = 5 induces the fragment {−2,−1, 0, +1, +2}.

In general:

card nat = K induces the fragment {0, . . . , K− 1}
card int = K induces the fragment {−dK/2e+ 1, . . . , +bK/2c}.

For free types, and often also for typedecl’d types, it usually makes sense
to specify cardinalities as a range of the form 1–n.

card = 〈int_seq〉 [1–6]

Specifies the default sequence of cardinalities to use for free types, type-
decl’d types (except bool), and the built-in types nat and int. This can be
overridden on a per-type basis using the card 〈type〉 option described above.

mult 〈const〉 = 〈int_seq〉
Specifies the sequence of multiplicities to use for a given datatype construc-
tor. The multiplicity of a constructor is the number of distinct values that
it can construct. Nonsensical values (e.g., mult Nil = 2) are repaired behind
the scenes. Most types declared using typedef are considered datatypes,
whose single constructor is the corresponding Abs_ function. The only ex-
ceptions are nat, int, × (product), and + (sum); the first two can be assigned
a cardinality using card, and the last two automatically get a cardinality
based on their arguments.

mult = 〈int_seq〉 [1–6]

Specifies the default sequence of multiplicities to use for datatype construc-
tors. This can be overridden on a per-constructor basis using the mult 〈const〉
option described above.

iter 〈const〉 = 〈int_seq〉
Specifies the sequence of iteration counts to use when unrolling a given in-
ductive predicate. By default, unrolling takes place for inductive predicates
that cannot be proved to be wellfounded, but this behavior is influenced by
the inductive_mood (§5.5) option. Internally, the iteration count is bounded
by the cardinality of the predicate’s domain.

iter = 〈int_seq〉 [1,2,4,8,16,32]

Specifies the sequence of iteration counts to use when unrolling inductive
predicates. This can be overridden on a per-predicate basis using the iter
〈const〉 option above.

lockstep = 〈bool〉 [false]

Specifies whether cardinalities of different types progress together or not.
Setting this option to true can dramatically reduce the number of scopes
tried, but it also diminishes the chances of finding a counterexample. Use
with care.

5.3 Output Format

verbose = 〈bool〉 [false]

Specifies whether Nitpick should explain what it does. This option is useful
to determine which scopes are tried or which SAT solver is used. The output
is clearer if batch_size (§5.5) is set to 1.

See also card (§5.2), mult (§5.2), and iter (§5.2).

show_skolems = 〈bool〉 [true]

Specifies whether the values of Skolem constants should be displayed as
part of counterexamples. Skolem constants correspond to bound variables

in the original formula and usually help understand why the counterexam-
ple falsifies the formula.

See also skolemize_depth (§5.5).

show_datatypes = 〈bool〉 [false]

Specifies whether the fragments used to approximate datatypes should be
displayed as part of counterexamples. Such fragments are sometimes help-
ful when investigating whether a potential counterexample is genuine or
spurious, but their potential for clutter is real.

show_consts = 〈bool〉 [false]

Specifies whether the values of constants associated with the original for-
mula (including its axioms) should be displayed as part of counterexam-
ples. These values are sometimes helpful when investigating whether a po-
tential counterexample is genuine or spurious, but their potential for clutter
is real.

max_potential = 〈int〉 [1]

Specifies the maximum number of potential counterexamples to display.
Setting this option to 0 speeds up the search for a genuine counterexample.

watch = 〈term_list〉
Specifies the list of terms whose values should be displayed as part of coun-
terexamples. This option suffers from an “observer effect”: Nitpick might
find different counterexamples for different values of this option.

See also check_potential (§5.4).

expect = 〈string〉
Specifies the expected outcome, which must be one of the following:

• sat: Nitpick found a genuine counterexample.

• weaksat: Nitpick found a potential counterexample.

• unsat: Nitpick found no counterexample.

• unknown: Nitpick encountered some problem (e.g., Kodkod ran out of
memory).

Nitpick emits an error if the actual outcome differs from the expected out-
come. This option is useful for regression testing.

5.4 Automatic Counterexample Checks

check_potential = 〈bool〉 [false]

Specifies whether potential counterexamples should be given to Isabelle’s

auto tactic to assess their validity. If a potential counterexample is shown to
be genuine, Nitpick displays a message to this effect and terminates.

See also max_potential (§5.3) and auto_timeout (§5.6).

check_genuine = 〈bool〉 [false]

Specifies whether genuine counterexamples should be given to Isabelle’s
auto tactic to assess their validity. If a “genuine” counterexample is shown
to be spurious, the user is kindly asked to send a bug report to the author
at blannospamchette@in.tum.de.

See also peephole (§5.5) and auto_timeout (§5.6).

5.5 Optimizations

sat_solver = 〈string〉 [auto]

Specifies which SAT solver to use. SAT solvers implemented in C or C++
tend to be faster than their Java counterparts, but they can be more difficult
to install. The supported solvers are listed below:

• SAT4J: SAT4J is a reasonably efficient solver written in Java. It is bun-
dled with Kodkodi and therefore easy to install. Do not attempt to
install the official SAT4J packages, because their API is incompatible
with Kodkod.

• SAT4JLight: Variant of SAT4J that is optimized for small problems.

• Jerusat: Jerusat 1.3 is an efficient solver written in C. To use Jerusat
with Nitpick, you must ensure that the Isabelle environment variable
JERUSAT_HOME is set to the directory that contains the Jerusat1.3 exe-
cutable. The C sources for Jerusat are available at http://www.cs.tau.
ac.il/~ale1/Jerusat1.3.tgz.

• BerkMin: BerkMin561 is an efficient solver written in C. To use Berk-
Min with Nitpick, you must ensure that the Isabelle environment vari-
able BERKMIN_HOME is set to the directory that contains the BerkMin561
executable. The BerkMin executables are available at http://eigold.
tripod.com/BerkMin.html.

• BerkMinAlloy: Variant of BerkMin that is included with Alloy 4 and
calls itself “sat56” in its banner text. To use this version of BerkMin
with Nitpick, you must ensure that the Isabelle environment variable
BERKMINALLOY_HOME is set to the directory that contains the berkmin
executable.

• RSat: RSat is an efficient solver written in C++. To use RSat, you must
ensure that the Isabelle environment variable RSAT_HOME is set to the
directory that contains the rsat executable. The C++ sources for RSat
are available at http://reasoning.cs.ucla.edu/rsat/. Nitpick has
been tested with version 2.01.

• zChaff : zChaff is an efficient solver written in C++. You can use the
version for Java bundled in nativesolver.tgz, which you will find on
Kodkod’s web site [3]. Alternatively, you can install a standard ver-
sion of zChaff and set the Isabelle environment variable ZCHAFF_HOME
to the directory that contains the zchaff executable. The C++ sources
and executables for zChaff are available at http://www.princeton.
edu/~chaff/zchaff.html. Nitpick has been tested with versions 2004-
05-13, 2004-11-15, and 2007-03-12.

• MiniSat: MiniSat is a minimalistic yet efficient solver written in C++.
You can use the version for Java bundled in nativesolver.tgz, which
you will find on Kodkod’s web site [3]. Alternatively, you can install a
standard version of MiniSat and set the Isabelle environment variable
MINISAT_HOME to the directory that contains the minisat executable.
The C++ sources and executables for MiniSat are available at http:
//minisat.se/MiniSat.html. Nitpick has been tested with versions
1.14 and 2.0 beta (2007-07-21).

• auto: If sat_solver is set to auto, Nitpick selects the first solver among
Jerusat, BerkMin, BerkMinAlloy, RSat, zChaff, and MiniSat that is con-
figured in Isabelle (i.e., whose _HOME variable is set). If none is con-
figured, it falls back on SAT4J, which should always be available. If
verbose = true, Nitpick displays which SAT solver was chosen.

batch_size = 〈int〉 [100]

Specifies the maximum number of Kodkod problems that should be lumped
together when invoking Kodkodi. Each problem corresponds to one scope.
Lumping problems together ensures that Kodkodi is launched less often.

See also verbose (§5.3).

normalize = 〈bool〉 [false]

Specifies whether the formula should be normalized-by-evaluation (as done
by Isabelle’s normal_form command) before Nitpick processes it further.
Normalization often transforms formulas in ways that make the search more
efficient; for example, it simplifies map f [x] to [ f x]. On the other hand, it
sometimes eliminates free variables and renames bound variable, making
the resulting counterexample more cryptic.

inductive_mood = 〈string〉 [realistic]

Specifies how to handle inductively defined predicates (and sets). The pos-
sible values are listed below:

• realistic: Try to prove that the inductive predicate is wellfounded us-
ing Isabelle’s lexicographic_order and sizechange tactics. If this succeeds,
use an efficient fixed point equation as specification of the predicate;
otherwise, unroll the predicates according to the iter 〈const〉 and iter
options (§5.2).

• optimistic: Tentatively treat all inductive predicates as if they were
wellfounded. This may lead to spurious counterexamples, but Kod-
kod is then invoked to check whether the example is genuine or not.
This approach is typically the most efficient, but it is also theoretically
the weakest. When Kodkod detects potentially non-wellfounded pred-
icates, it emits a message listing the offending predicates and suggest-
ing to try again with inductive_mood set to realistic or pessimistic.

• pessimistic: Treat all inductive predicates as if they were not well-
founded. The predicates are then unrolled according to the iter 〈const〉
and iter options (§5.2).

See also wellfounded_timeout (§5.6).

specialize_funs = 〈bool〉 [true]

Specifies whether functions invoked with fixed arguments should be spe-
cialized behind the scenes. This optimization often reduces the search space
drastically, especially for higher-order functions.

See also show_consts (§5.3).

skolem_depth = 〈int〉 [4]

Specifies the maximum depth at which Skolemization takes place, expressed
as the number of outer quantifiers. The value −1 disables skolemization,
0 means that only the outermost ∀-quantifiers (or negated ∃-quantifiers) in
the original (unnegated) formula are skolemized, and n > 0 means that
∀-quantifiers within the scope of at most n ∃-quantifiers are skolemized.
However, for performance reasons, ∀-quantifiers in the scope of a higher-
order ∃-quantifier are normally left unchanged.

peephole = 〈bool〉 [true]

Specifies whether Nitpick should perform peephole optimization of the gen-
erated Kodkod formulas. These optimizations can make a significant differ-
ence. Unless you are tracking down a bug in Nitpick or distrust the peep-
hole optimizer, you should leave this option enabled.

sym_break = 〈int〉 [20]

Specifies an upper bound on the number of relations for which Kodkod
generates symmetry breaking predicates. According to the Kodkod docu-
mentation [4], “in general, the higher this value, the more symmetries will
be broken, and the faster the formula will be solved. But, setting the value
too high may have the opposite effect and slow down the solving.” The
default of 20 is the same as in Kodkod and Alloy.

sharing_depth = 〈int〉 [3]

Specifies the depth to which Kodkod should check circuits for equivalence
during the translation to SAT. The default of 3 is the same as in Alloy. The
minimum allowed depth is 1. Increasing the sharing may result in a smaller
SAT problem, but can also slow down Kodkod.

flatten_props = 〈bool〉 [false]

Specifies whether Kodkod should try to eliminate intermediate Boolean
variables. Although this might sound like a good idea, in practice it can
drastically slow down Kodkod and is generally not recommended.

5.6 Timeouts

timeout = 〈time〉 [30 s]

Specifies the maximum amount of time that Nitpick should use to find a
counterexample. Nitpick tries to honor this constraint as well as it can but
offers no guarantees.

auto_timeout = 〈time〉 [100 ms]

Specifies the maximum amount of time that auto should use when checking
a counterexample. Nitpick tries to honor this constraint as well as it can but
offers no guarantees.

See also check_potential (§5.4) and check_genuine (§5.4).

wellfounded_timeout = 〈time〉 [200 ms]

Specifies the maximum amount of time that lexicographic_order and size-
change should use when checking whether an inductive predicate is well-
founded. Nitpick tries to honor this constraint as well as it can but offers no
guarantees.

See also inductive_mood (§5.5).

6 Known Bugs and Limitations

Here are the known bugs and limitations in Nitpick at the time of writing:

• User-defined axioms might be ignored or mishandled by Nitpick.

• For functions defined using primrec, fun, or function, Nitpick sometimes
generates spurious “genuine” counterexamples about theorems that refer
to values for which the function is not defined. For example:

primrec prec where
“prec (Suc n) = n”

lemma “prec 0 = undefined”
nitpick

Nitpick found a genuine counterexample for card nat = 2:

Trivial

by (auto simp: prec_def )

Such theorems are considered bad style because they rely on the internal
representation of functions synthesized by Isabelle, which is an implemen-
tation detail.

• Coinduction is handled by expanding the greatest fixed point definition,
which slows down Kodkod horribly. Future versions of Nitpick are ex-
pected to unroll coinductive definitions.

Comments and bug reports concerning Nitpick or this manual should be directed
to blannospamchette@in.tum.de.

